

INFOAFP – Exam

Andres Löh

Wednesday, 16 April 2008, 09:00–12:00

Evaluation strategies (19 points total)

1 (5 points). Give an example of a Haskell expression of type *Bool* that evaluates to *True* and that would not terminate (i.e., loop forever) in a language with strict evaluation.

Using equational reasoning, give the reduction sequence of your expression to *True* and indicate clearly where a strict reduction strategy would select another redex. •

2 (4 points). Haskell has non-strict semantics (i.e., the implementations use lazy evaluation) and is called a pure language. (S)ML on the other hand has strict semantics and is an impure language.

What does purity mean in this context? Give an (small!) example of why Haskell is considered pure and (S)ML is not. [Syntactic correctness, particularly of (S)ML code, is not important.] •

3 (4 points). Would a lazy impure language or a strict pure language be possible? Would such programming languages be useful? Discuss briefly. •

4 (6 points). Consider the following Haskell functions. Divide the functions into equivalence classes, i.e., group the functions that are semantically equivalent (efficiency is irrelevant). Give as many examples as needed to demonstrate that each of the classes has indeed different behaviour.

```
s1 :: [a] -> b -> b
s1 xs y = case xs of { [] -> y; _ -> y }

s2 :: [a] -> b -> b
s2 xs y = seq xs y

s3 :: [a] -> b -> b
s3 xs y = y

s4 :: [a] -> b -> b
s4 xs y = if null xs then y else y

s5 :: [a] -> b -> b
s5 xs y = if map (const 0) xs == [] then y else y

s6 :: [a] -> b -> b
s6 xs y = case xs of { [] -> y; [x] -> y; _ -> y }

s7 :: [a] -> b -> b
s7 xs y = seq [xs] y
```

Interactive programs (12 points total)

Consider the following datatype:

```
data GP a = End a
          | Get (Int → GP a)
          | Put Int (GP a)
```

A value of type *GP* can be used to describe programs that read and write integer values and return a final result of type *a*. Such a program can end immediately (*End*). If it reads an integer, the rest of the program is described as a function depending on this integer (*Get*). If the program writes an integer (*Put*), the value of that integer and the rest of the program are recorded.

The following expression describes a program that continuously reads integers and prints them:

```
echo = Get (λn → Put n echo)
```

5 (1 point). What is the (inferred) type of *echo*?

6 (4 points). Write a function

```
run :: GP a → IO a
```

that can run a *GP*-program in the *IO* monad. A *Get* should read an integer from the console, and *Put* should write an integer to the console.

Here is an example run from GHCi:

```
Main> run echo
? 42
42
? 28
28
? 1
1
? -5
-5
? Interrupted.
Main>
```

[To better distinguish inputs from outputs, this version of *run* prints a question mark when expecting an input. It is not required that your version does the same.] •

7 (3 points). Write a *GP*-program *add* that reads two integers, writes the sum of the two integers, and ultimately returns (). •

8 (4 points). Write a *GP*-program *accum* that reads an integer. If the integer is 0, it returns the current total. If the integer is not 0, it adds the integer to the current total, prints the current total, and starts from the beginning. •

Simulation (21 points total)

9 (4 points). Instead of running a *GP*-program in the *IO* monad, we can also simulate the behaviour of such a program by providing a (possibly infinite) list of input values. Write a function

$$\text{simulate} :: GP\ a \rightarrow [Int] \rightarrow (a, [Int])$$

that takes such a list of input values and returns the final result plus the (possibly infinite) list of all the output values generated. •

10 (3 points). What is the result of evaluating the following two expressions?

$$\text{simulate accum } [5,4..0]$$
$$\text{simulate accum } [5,4..1]$$

11 (4 points). Define a QuickCheck property that states the following property using *simulate*:

“If *echo* is given *n* numbers as input, then the first *n* numbers of its output will be identical to the input.” •

12 (4 points). Which parts of the definition of *simulate* are covered by your property, and which are not? (I.e., which parts of the definition of *simulate* would be highlighted by HPC after running QuickCheck on your property – assuming that QuickCheck generates suitably random lists.) •

13 (6 points). This is an attempt to define a QuickCheck property for *accum*:

$$\begin{aligned} \text{accumP} &:: [Int] \rightarrow \text{Property} \\ \text{accumP } xs &= \text{all } (\lambda x \rightarrow x > 0) xs \implies \\ &\quad \text{simulate accum } (xs ++ [0]) == (\text{last } sl, sl) \\ &\quad \textbf{where } sl = \text{scanl1 } (+) xs \end{aligned}$$

Here, *scanl1* is defined as follows

$$\begin{aligned} \text{scanl1} &:: (a \rightarrow a \rightarrow a) \rightarrow [a] \rightarrow [a] \\ \text{scanl1 } f [] &= [] \\ \text{scanl1 } f (x : xs) &= \text{scanl } f\ x\ xs \\ \\ \text{scanl} &:: (a \rightarrow b \rightarrow a) \rightarrow a \rightarrow [b] \rightarrow [a] \\ \text{scanl } f\ x\ xs &= x : \textbf{case } xs \textbf{ of} \\ &\quad [] \rightarrow [] \\ &\quad y : ys \rightarrow \text{scanl } f\ (f\ x\ y)\ ys \end{aligned}$$

There are at least two problems with this property. Describe how they can be fixed [a description is sufficient]. •

Functors and monads (24 points total)

A map function for GP can be defined as follows:

```
instance Functor GP where
  fmap f (End x) = End (f x)
  fmap f (Get g) = Get (fmap f o g)
  fmap f (Put n x) = Put n (fmap f x)
```

14 (2 points). Describe the difference between the behaviour of *run accum* and the behaviour of *run (fmap (*2) accum)*. •

15 (8 points). Prove the **first** of the two laws using equational reasoning (and ignoring that values can be \perp).

Note that if you want to prove a property $P p$ for any $p :: GP a$ via structural induction, you have to prove the following three cases:

```
∀x.    P (End x)
∀g.    (∀x.P (g x)) ⇒ P (Get g)
∀n p.  P p ⇒ P (Put n p)
```

(Here, \Rightarrow denotes logical implication.) Note that the second case is slightly unusual due to the function argument of *Get*: you may assume that $P (g x)$ holds for any value of x ! •

16 (5 points). Define a sensible monad instance for GP . •

17 (5 points). Define a sensible *MonadState* instance for GP . Recall the *MonadState* class:

```
class (Monad m) ⇒ MonadState s m | m → s where
  get :: m s
  put :: s → m ()
```

18 (4 points). What is the difference between the normal state monad as defined in module *Control.Monad.State* and GP ? Discuss whether you think it is a good idea to make GP an instance of *MonadState*. •

Type classes (10 points total, 5 bonus points)

19 (2 points). Consider this program:

```
equal :: (Eq s, MonadState s m) ⇒ m Bool
equal = do
  x ← get
  y ← get
  return (x == y)
```

Is the given type signature the most general type signature for *equal*? What would happen if the type signature would be omitted? •

20 (8 points). Translate type classes into explicit evidence in the above function *equal*. Desugar the **do**-notation in the process [use the “simple” desugaring, without the possibility to pattern match on the left hand side of an arrow]. Define the dictionary types that are required – you may omit class methods that are not relevant to this example. You may also declare local abbreviations using **let**. •

21 (5 *bonus points*). Haskell does not offer a scoping mechanism for instances. Instances are always exported from modules, even if nothing else is. Also, instances cannot be local. For example,

```
let instance Eq Char where
    x == y = ord (toUpper x) == ord (toUpper y)
in "hello" == "HeLlO"
```

(using *ord* and *toUpper* from *Data.Char*) is not legal Haskell.

Why do you think this decision has been made? Are there any problems you can think of? •

GADTs and kinds (14 points total)

Here is a variation of *GP*:

```
data GP' :: * -> * where
    Return :: a -> GP' a
    Bind   :: GP' a -> (a -> GP' b) -> GP' b
    Get'   :: GP' Int
    Put'   :: Int -> GP' ()
```

This is a GADT. The type *GP'* can trivially be made an instance of the classes *Monad* and *MonadState*:

```
instance Monad GP' where
    return = Return
    (>>=) = Bind

instance MonadState Int GP' where
    get = Get'
    put = Put'
```

22 (6 points). A value of type *GP* can easily be transformed into a value of type *GP'* as follows:

```
gp2gp' :: GP a -> GP' a
gp2gp' (End x) = Return x
gp2gp' (Get f) = Get' >>= \x -> gp2gp' (f x)
gp2gp' (Put n k) = Put' n >> gp2gp' k
```

Define a transformation in the other direction, i.e., a function

$$gp'2gp :: GP' a \rightarrow GP a$$

such that $gp'2gp \circ gp2gp' \equiv id$ for all values that do not contain \perp . Does $gp2gp' \circ gp'2gp$ also yield the identity? [No formal proof is required.] •

23 (4 points). Do the monad laws hold for GP and GP' ? [Give a counterexample if not, argue briefly if yes – no formal proof is required.]

Describe advantages and disadvantages of the two variants. •

24 (4 points). Define type synonyms of kind

$$((* \rightarrow *) \rightarrow *) \rightarrow *$$

and

$$(* \rightarrow *) \rightarrow (* \rightarrow *) \rightarrow (* \rightarrow *)$$

without using any user-defined **datatypes**. •